
Razonamiento por defecto

x

Ejemplo de razonamiento por defecto

El animal_1 es un p�ajaro

Normalmente, los p�ajaros vuelan.

Por tanto, el animal_1 vuela.

x

Programa P1

u

Progama P1

pajaro(animal_1).

vuela(X) :-

pajaro(X),

normal(X).

u

Modelos del programa P1

{pajaro(animal_1)}

{pajaro(animal_1), vuela(animal_1)}

{pajaro(animal_1), normal(animal_1), vuela(animal_1)}

u

Consecuencia

P1 |=/= vuela(animal_1)

Razonamiento por defecto

x

Programa P2 con anormal/1

u

Programa P2

:- dynamic anormal/1.

pajaro(animal_1).

vuela(X) :-

pajaro(X),

not anormal(X).

u

Sesi�on

?- vuela(animal_1).

Yes

u

Traza

?- vuela(animal_1).

Call: ( 7) vuela(animal_1) ?

Call: ( 8) pajaro(animal_1) ?

Exit: ( 8) pajaro(animal_1) ?

^ Call: ( 8) not anormal(animal_1) ?

Call: ( 9) anormal(animal_1) ?

Fail: ( 9) anormal(animal_1) ?

^ Exit: ( 8) not anormal(animal_1) ?

Exit: ( 7) vuela(animal_1) ?

Yes

Razonamiento por defecto

x

Extensi�on del conocimiento

u

Nuevo conocimiento

El animal_1 es un avestruz.

Los avestruces son p�ajaros que no vuelan.

u

Programa extendido

:- dynamic anormal/1.

pajaro(animal_1).

avestruz(animal_1).

vuela(X) :-

pajaro(X),

not anormal(X).

anormal(X) :- avestruz(X).

u

Traza

?- vuela(animal_1).

Call: ( 7) vuela(animal_1) ?

Call: ( 8) pajaro(animal_1) ?

Exit: ( 8) pajaro(animal_1) ?

^ Call: ( 8) not anormal(animal_1) ?

Call: ( 9) anormal(animal_1) ?

Call: ( 10) avestruz(animal_1) ?

Exit: ( 10) avestruz(animal_1) ?

Exit: ( 9) anormal(animal_1) ?

^ Fail: ( 8) not anormal(animal_1) ?

Fail: ( 7) vuela(animal_1) ?

No

Razonamiento por defecto

x

Cancelaci�on reglas por defectos mediante reglas

espec���cas

u

Regla por defecto: \Normalmente, los p�ajaros vue-

lan"

u

Regla espec���ca: \Los avestruces no vuelan"

x

Razonamiento mon�otono y no{mon�otono

u

Razonamiento mon�otono

P1 |= C y P2 extiende a P1, entonces P2 |= C.

u

Razonamiento mon�otono

P1 |= C y P2 extiende a P1, es posible P2 |=/= C.

Razonamiento por defecto

x

Programa con reglas y reglas con excepciones

(defectos)

u

Programa objeto

defecto((vuela(X) :- pajaro(X))).

regla((pajaro(X) :- avestruz(X))).

regla((not vuela(X) :- avestruz(X))).

regla((avestruz(animal_1) :- true)).

regla((pajaro(animal_2) :- true)).

u

Sesi�on

?- explica(vuela(X),E).

X = animal_2

E = [defecto((vuela(animal_2) :- pajaro(animal_2))),

regla((pajaro(animal_2) :- true))] ;

No

?- explica(not vuela(X),E).

X = animal_1

E = [regla((not vuela(animal_1) :- avestruz(animal_1))),

regla((avestruz(animal_1) :- true))] ;

No

?- explica(vuela(animal_2),_).

Yes

?- explica(vuela(animal_1),_).

No

Razonamiento por defecto

u

Metaprograma para explicaciones

explica(A,E) :-

explica(A,[],E).

explica(true,E,E) :- !.

explica((A,B),E0,E) :- !,

explica(A,E0,E1),

explica(B,E1,E).

explica(A,E0,E) :-

prueba(A,E0,E).

explica(A,E0,[defecto((A:-B))|E]) :-

defecto((A:-B)),

explica(B,E0,E),

not contradiccion(A,E).

prueba(true,E,E) :- !.

prueba((A,B),E0,E) :- !,

prueba(A,E0,E1),

prueba(B,E1,E).

prueba(A,E0,[regla((A:-B))|E]) :-

regla((A:-B)),

prueba(B,E0,E).

contradiccion(not A,E) :- !,

prueba(A,E,_E1).

contradiccion(A,E) :-

prueba(not A,E,_E1).

Razonamiento por defecto

x

Explicaciones de hechos contradictorios

u

Programa objeto

defecto((not vuela(X) :- mamifero(X))).

defecto((vuela(X) :- vampiro(X))).

defecto((not vuela(X) :- muerto(X))).

regla((mamifero(X) :- vampiro(X))).

regla((vampiro(dracula) :- true)).

regla((muerto(dracula) :- true)).

u

Sesi�on

:- [p5].

?- explica(vuela(dracula),E).

E = [defecto((vuela(dracula) :- vampiro(dracula))),

regla((vampiro(dracula) :- true))] ;

No

?- explica(not vuela(dracula),E).

E = [defecto((not vuela(dracula) :- mamifero(dracula))),

regla((mamifero(dracula) :- vampiro(dracula))),

regla((vampiro(dracula) :- true))] ;

E = [defecto((not vuela(dracula) :- muerto(dracula))),

regla((muerto(dracula) :- true))] ;

No

Razonamiento por defecto

x

Cancelaci�on entre defectos mediante nombres

u

Programa objeto

defecto(mamiferos_no_vuelan(X),

(not vuela(X) :- mamifero(X))).

defecto(vampiros_vuelan(X),

(vuela(X) :- vampiro(X))).

defecto(muertos_no_vuelan(X),

(not vuela(X) :- muerto(X))).

regla((mamifero(X) :- vampiro(X))).

regla((vampiro(dracula) :- true)).

regla((muerto(dracula) :- true)).

regla((not mamiferos_no_vuelan(X) :- vampiro(X))).

regla((not vampiros_vuelan(X) :- muerto(X))).

u

Modi�caci�on de explica

explica(A,E0,[defecto(Nombre)|E]) :-

defecto(Nombre,(A:-B)),

explica(B,E0,E),

not contradiccion(Nombre,E),

not contradiccion(A,E).

u

Sesi�on

?- explica(vuela(dracula),E).

No

?- explica(not vuela(dracula),E).

E = [defecto(muertos_no_vuelan(dracula)),

regla((muerto(dracula) :- true))]

Yes

Razonamiento abductivo

x

Problema de la abducci�on

Dados P un programa l�ogico y

O una observaci�on (un hecho b�asico

en el lenguaje de P)

Encontrar E una explicaci�on (una lista de hechos at�omicos

en el lenguaje de P cuyos predicados no son

cabezas de cl�ausulas de P) tal que

P U E |- O)

x

Abducci�on para programas de�nidos

u

Programa objeto

europeo(X) <- espa~nol(X).

espa~nol(X) <- andaluz(X).

europeo(X) <- italiano(X).

u

Sesi�on

?- abduce(europeo(juan),E).

E = [andaluz(juan)] ;

E = [italiano(juan)] ;

No

Razonamiento abductivo

u

Programa

:- op(1200,xfx,<-).

abduce(O,E) :-

abduce(O,[],E).

abduce(true,E,E) :- !.

abduce((A,B),E0,E) :- !,

abduce(A,E0,E1),

abduce(B,E1,E).

abduce(A,E0,E) :-

(A <- B),

abduce(B,E0,E).

abduce(A,E,E) :-

member(A,E).

abduce(A,E,[A|E]) :-

not member(A,E),

abducible(A).

abducible(A) :-

not (A <- _B).

Razonamiento abductivo

x

Problemas al aplicar abducci�on a programas in-

de�nidos

u

Programa objeto

vuela(X) <- pajaro(X), not anormal(X).

anormal(X) <- avestruz(X).

pajaro(X) <- avestruz(X).

pajaro(X) <- palomo(X).

u

Sesi�on

?- abduce(vuela(animal_1),E).

E = [not anormal(animal_1), avestruz(animal_1)] ;

E = [not anormal(animal_1), palomo(animal_1)] ;

No

u

Problemas:

* Explicaci�on contradictoria

* Explicaci�on con predicado no abducible

Razonamiento abductivo

x

Abducci�on para programas generales

u

Metaprograma abductivo

:- op(1200,xfx,<-).

abduce(O,E) :-

abduce(O,[],E).

abduce(true,E,E) :- !.

abduce((A,B),E0,E) :- !,

abduce(A,E0,E1),

abduce(B,E1,E).

abduce(A,E0,E) :-

(A <- B),

abduce(B,E0,E).

abduce(A,E,E) :-

member(A,E).

abduce(A,E,[A|E]) :-

not member(A,E),

abducible(A),

not abduce_not(A,E,E).

abduce(not(A),E0,E) :-

not member(A,E0),

abduce_not(A,E0,E).

Razonamiento abductivo

abduce_not((A,B),E0,E) :- !,

abduce_not(A,E0,E);

abduce_not(B,E0,E).

abduce_not(A,E0,E) :-

setof(B,(A <- B),L),

abduce_not_l(L,E0,E).

abduce_not(A,E,E) :-

member(not(A),E).

abduce_not(A,E,[not(A)|E]) :-

not member(not(A),E),

abducible(A),

not abduce(A,E,E).

abduce_not(not(A),E0,E) :-

not member(not(A),E0),

abduce(A,E0,E).

abduce_not_l([],E,E).

abduce_not_l([B|R],E0,E) :-

abduce_not(B,E0,E1),

abduce_not_l(R,E1,E).

abducible(A) :-

A \= not(_X),

not (A <- _B).

u

Sesi�on con el programa objeto anterior

?- abduce(vuela(animal_1),E).

E = [not avestruz(animal_1), palomo(animal_1)]

Yes

Diagn�ostico mediante abducci�on

x

Representaci�on de un sumador

+----+

X ----+->| |S

| |xor1|--+--+

Y --+-|->| | | | +----+

| | +----+ | +------------>| |

| | | | |xor2|---Suma

Z --|-|-------+--|--------------->| |

| | | | +----+ +----+

| +-------|--|-->| |A1

| | | |and1|---+

+---------|--|-->| | | +----+

| | +----+ +-->| |

| | |or1 |---Acarreo

| | +----+ +-->| |

| +-->| |A2 | +----+

| |and1|---+

+----->| |

+----+

Diagn�ostico mediante abducci�on

x

De�nici�on del sumador

sumador(X,Y,Z,Acarreo,Suma) :-

xor(X,Y,S),

xor(Z,S,Suma),

and(X,Y,A1),

and(Z,S,A2),

or(A1,A2,Acarreo).

and(1,1,1). and(1,0,0).

and(0,1,0). and(0,0,0).

or(1,1,1). or(1,0,1).

or(0,1,1). or(0,0,0).

xor(1,0,1). xor(0,1,1).

xor(1,1,0). xor(0,0,0).

tabla :-

format('X Y Z A S~n'),

tabla2.

tabla2 :-

member(X,[0,1]),

member(Y,[0,1]),

member(Z,[0,1]),

sumador(X,Y,Z,A,S),

format('~a ~a ~a ~a ~a~n',[X,Y,Z,A,S]),

fail.

tabla2.

Diagn�ostico mediante abducci�on

x

Sesi�on con el sumador

?- tabla.

X Y Z A S

0 0 0 0 0

0 0 1 0 1

0 1 0 0 1

0 1 1 1 0

1 0 0 0 1

1 0 1 1 0

1 1 0 1 0

1 1 1 1 1

Yes

x

Modelo de fallo del sumador

sumador(X,Y,Z,Acarreo,Suma) <-

xorg(xor1,X,Y,S),

xorg(xor2,Z,S,Suma),

andg(and1,X,Y,A1),

andg(and2,Z,S,A2),

org(or1,A1,A2,Acarreo).

xorg(_N,X,Y,Z) <- xor(X,Y,Z).

xorg(N,1,1,1) <- fallo(N=f1).

xorg(N,0,0,1) <- fallo(N=f1).

xorg(N,1,0,0) <- fallo(N=f0).

xorg(N,0,1,0) <- fallo(N=f0).

Diagn�ostico mediante abducci�on

andg(_N,X,Y,Z) <- and(X,Y,Z).

andg(N,0,0,1) <- fallo(N=f1).

andg(N,1,0,1) <- fallo(N=f1).

andg(N,0,1,1) <- fallo(N=f1).

andg(N,1,1,0) <- fallo(N=f0).

org(_N,X,Y,Z) <- or(X,Y,Z).

org(N,0,0,1) <- fallo(N=f1).

org(N,1,0,0) <- fallo(N=f0).

org(N,0,1,0) <- fallo(N=f0).

org(N,1,1,0) <- fallo(N=f0).

x

Diagn�ostico mediante abducci�on

diagnostico(Observacion,Diagnostico) :-

abduce(Observacion,Diagnostico).

:- abolish(abducible,2).

abducible(fallo(_X)).

x

Sesi�on de diagn�ostico

?- diagnostico(sumador(0,0,1,1,0),D).

D = [fallo(or1 = f1), fallo(xor2 = f0)] ;

D = [fallo(and2 = f1), fallo(xor2 = f0)] ;

D = [fallo(and1 = f1), fallo(xor2 = f0)] ;

D = [fallo(and2 = f1), fallo(and1 = f1), fallo(xor2 = f0)] ;

D = [fallo(xor1 = f1)] ;

D = [fallo(or1 = f1), fallo(and2 = f0), fallo(xor1 = f1)] ;

D = [fallo(and1 = f1), fallo(xor1 = f1)] ;

D = [fallo(and2 = f0), fallo(and1 = f1), fallo(xor1 = f1)] ;

No

Diagn�ostico mediante abducci�on

x

Diagn�ostico m��nimo

diagnostico_minimo(O,D) :-

diagnostico(O,D),

not((diagnostico(O,D1),

subconjunto_propio(D1,D))).

subconjunto_propio([],Ys):-

Ys \= [].

subconjunto_propio([X|Xs],Ys):-

select(Ys,X,Ys1),

subconjunto_propio(Xs,Ys1).

x

Diagn�ostico m��nimo del sumador

?- diagnostico_minimo(sumador(0,0,1,1,0),D).

D = [fallo(or1 = f1), fallo(xor2 = f0)] ;

D = [fallo(and2 = f1), fallo(xor2 = f0)] ;

D = [fallo(and1 = f1), fallo(xor2 = f0)] ;

D = [fallo(xor1 = f1)] ;

No

Defectos mediante abducci�on

x

Traducci�on del programa objeto

no_vuela(X) <- mamifero(X), not mamifero_volador(X).

vuela(X) <- vampiro(X), not vampiro_no_volador(X).

no_vuela(X) <- muerto(X), not muerto_volador(X).

mamifero(X) <- vampiro(X).

vampiro(dracula) <- true.

muerto(dracula) <- true.

mamifero_volador(X) <- vampiro(X).

vampiro_no_volador(X) <- muerto(X).

:- abolish(abducible,1).

abducible(mamifero_volador(_)).

abducible(vampiro_no_volador(_)).

abducible(muerto_volador(_)).

x

Sesi�on

?- abduce(vuela(X),E).

No

?- abduce(no_vuela(X),E).

X = dracula

E = [not muerto_volador(dracula)] ;

No

